unittest
{
	assert( [ "2", "3 1 4 1 5 9" ].parse.expand.solve == 1 );
	assert( [ "1", "1 2 3" ].parse.expand.solve == -1 );
	assert( [ "3", "8 2 2 7 4 6 5 3 8" ].parse.expand.solve == 5 );
}



import std.algorithm;
import std.container;
import std.conv;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto n = input.front.to!( long );
	input.popFront;
	auto as = input.front.split.to!( long[] );
	return tuple( n, as );
}

auto solve( long n, long[] as )
{
	// 「前1/3のグループ」と「後ろ2/3のグループ」と「後ろのグループから取り除かれたグループ」に分ける
	auto rbt_a = redBlackTree!( true )( as[ 0 .. n ] );
	auto rbt_b = redBlackTree!( true )( as[ n .. $ ] );
	auto rbt_c = redBlackTree!( true, long )();
	while( n < rbt_b.length )
	{
		rbt_c.insert( rbt_b.back );
		rbt_b.removeBack;
	}
	
	// 後ろのグループから一つずつ前のグループに移動したときのスコアを計算
	auto at = rbt_a[].sum;
	auto bt = rbt_b[].sum;
	auto ans = at - bt;
	foreach( a; as[ n .. n * 2 ] )
	{
		if( a in rbt_c )
		{
			rbt_c.removeKey( a );
		}
		else
		{
			rbt_b.insert( rbt_c.front );
			rbt_b.removeKey( a );
			bt += rbt_c.front - a;
			rbt_c.removeFront;
		}
		rbt_a.insert( a );
		at += a - rbt_a.front;
		rbt_a.removeFront;
		
		ans = max( ans, at - bt );
	}
	return ans;
}
